

## Content of Use a Search to Move Similar Active Directory Objects.vbs ( Site 1 )

' Description: Searches for all user account objects whose department attribute is Human Resources. The script then moves the user accounts that are not already in the HR OU to this OU.

```
Set objConnection = CreateObject("ADODB.Connection")
objConnection.Open "Provider=ADsDSOObject;"

Set objCommand = CreateObject("ADODB.Command")
objCommand.ActiveConnection = objConnection

objCommand.CommandText = _
    "<LDAP://dc=NA,dc=fabrikam,dc=com>;" & _
    "(&(&(objectCategory=person)(objectClass=user) & _" & _
    "(department=Human Resources));" & _
    "ADsPath,distinguishedName, name; subtree"

Set objRecordSet = objCommand.Execute
Set objOU = GetObject("LDAP://ou=HR,dc=NA,dc=fabrikam,dc=com")

Do Until objRecordSet.EOF
    strADsPath = objRecordSet.Fields("ADsPath")

    strDNRecord=lcase(objRecordSet.Fields("distinguishedName"))
    strDNCompare=lcase("cn=" & objRecordSet.Fields("name") & _
        ",ou=HR,dc=NA,dc=fabrikam,dc=com")

    If strDNRecord <> strDNCompare Then
        objOU.MoveHere strADsPath, vbNullString
        WScript.Echo objRecordSet.Fields("distinguishedName") & " Moved."
    Else
        WScript.Echo objRecordSet.Fields("distinguishedName") & " Not Moved."
    End If
    objRecordSet.MoveNext
Loop

objConnection.Close
```